This is a Maple worksheet for doing the calculations
for the Leslie population distribution model for MATH 240.
Michael Monagan, March, 2016.

For a population with 3 age groups with fertility rates F1, F2, F3 and
survival rates P1, P2, P3 the 3 by 3 Leslie matrix L looks like this.

```
> L := Matrix([[F1,F2,F3],[P1,0,0],[0,P2,P3]]);
```

$$L := \begin{bmatrix} F1 & F2 & F3 \\ P1 & 0 & 0 \\ 0 & P2 & P3 \end{bmatrix}$$

The data in the 3 by 3 Leslie matrix below correspondes to a seal population
on Sable island which is an island off the coast of Nova Scotia. The three age groups
are seal pups (0-4yrs), young adult seals (4-8yrs) and mature adult seals (8+ yrs).

```
> (F1,F2,F3,P1,P2,P3) := (0.0,1.221,2.0,0.597,0.808,0.808):
> L := Matrix([[F1,F2,F3],[P1,0,0],[0,P2,P3]]);
```

$$L := \begin{bmatrix} 0. & 1.221 & 2.0 \\ 0.597 & 0 & 0 \\ 0 & 0.808 & 0.808 \end{bmatrix}$$

Let's run the model for 20 time periods starting with initial population vector [1, 0, 0].

```
> N[0] := Vector([1.0,0.0,0.0]):
  for i from 1 to 20 do N[i] := L.N[i-1]; od:
```

Let's see what happens between N[5], N[10], and N[20]

```
> N[5], N[10], N[20];
```

$$\begin{bmatrix} 2.03633870697600 \\ 0.782588653283493 \\ 1.00394370320621 \end{bmatrix}, \begin{bmatrix} 14.0399769482490 \\ 5.68204043761159 \\ 6.88540964282670 \end{bmatrix}, \begin{bmatrix} 684.107775948775 \\ 276.900111619070 \\ 335.463042196983 \end{bmatrix}$$

The population is increasing rapidly!

Lets scale the vectors to get the population **distribution** vectors

```
> for i from 1 to 20 do Dist[i] := N[i]/add(N[i][j],j=1..3); od:
> Dist[5], Dist[19], Dist[20];
```

$$\begin{bmatrix} 0.532672609975476 \\ 0.204712280453953 \\ 0.262615109570570 \end{bmatrix}, \begin{bmatrix} 0.527669200199894 \\ 0.213579882062193 \\ 0.258750917737912 \end{bmatrix}, \begin{bmatrix} 0.527669198161557 \\ 0.213579884640605 \\ 0.258750917197838 \end{bmatrix}$$

Oberserve that the population **distribution** has stabililized, but the population is still
increasing by a factor of N[20] / N[19]

Let us see how much the population of each age group is increasing after the distribution has stabilized.

```
> N[20], N[19], <seq(N[20][i]/N[19][i],i=1..3)>
```

$$\begin{bmatrix} 684.107775948775 \\ 276.900111619070 \\ 335.463042196983 \end{bmatrix}, \begin{bmatrix} 463.819282444003 \\ 187.735929262185 \\ 227.441103159823 \end{bmatrix}, \begin{bmatrix} 1.47494466453401 \\ 1.47494468803764 \\ 1.47494466715303 \end{bmatrix}$$

This means $\lambda = 1.475$ is an eigenvalue of L with eigenvector D[20] = [0.528, 0.213, 0.259].
Let us confirm this by calculating the eigenvalues of L directly

```
> LinearAlgebra[Eigenvalues](L);
```

$$\begin{bmatrix} -0.333472336208076 + 0.378900636526870\,\mathrm{I} \\ -0.333472336208076 - 0.378900636526870\,\mathrm{I} \\ 1.47494467241615 + 0.\,\mathrm{I} \end{bmatrix}$$

Two complex eigenvalues and one real positive eigenvalue. Note, Maple uses I for the complex unit instead of i .

To investigate what would happen if the survival probabilities P1, P2, and P3 were halved.

```
> P1,P2,P3 := 0.597/2, 0.808/2, 0.808/2;
```

$$P1, P2, P3 := 0.2985000000, 0.4040000000, 0.4040000000$$

```
> L := Matrix([[F1,F2,F3],[P1,0,0],[0,P2,P3]]);
```

$$L := \begin{bmatrix} 0. & 1.221 & 2.0 \\ 0.2985000000 & 0 & 0 \\ 0 & 0.4040000000 & 0.4040000000 \end{bmatrix}$$

```
> LinearAlgebra[Eigenvalues](L);
```

$$\begin{bmatrix} 0.914722455035491 + 0.\,\mathrm{I} \\ -0.255361227517745 + 0.193627067819679\,\mathrm{I} \\ -0.255361227517745 - 0.193627067819679\,\mathrm{I} \end{bmatrix}$$

The eigenvalue $\lambda = 0.914$ is less than 0 which means the population will die out.
To investigate this let's start with a large population of [100, 100, 100] and see what happens.

```
> N[0] := Vector([100.0,100.0,100.0]):
  for i to 20 do N[i] := L.N[i-1]; od:
> N[5], N[10], N[20];
```

$$\begin{bmatrix} 169.116560757654 \\ 55.5174758938437 \\ 43.8752468079673 \end{bmatrix}, \begin{bmatrix} 108.696983074759 \\ 35.4723716930104 \\ 28.0594184853667 \end{bmatrix}, \begin{bmatrix} 44.5779473607698 \\ 14.5470543939876 \\ 11.5072480364767 \end{bmatrix}$$